
C:\jim\COURSES\8858\code-bk 2012\M4-2.gms  Monday, January 09, 2012 4:51:12 AM Page 1

$TITLE: M4-2.GMS: Find all pure-strategy Nash equilibrium
*                 with discrete strategy sets

$ONTEXT
two firms, one in country h and one in country f
each firm chooses one of three strategies:
 don't enter: strategy 0
 enter with a single plant and export to the other country: strategy 1
 enter with plants in both countries (horizontal multinational)
   strategy 2
$OFFTEXT

SETS R  strategies for firm h /SH0, SH1, SH2/
     C  strategies for firm f /SF0, SF1, SF2/;

ALIAS(R,RR)
ALIAS(C,CC);

PARAMETERS
 ROWMAX(R,C)    maximum value over the rows for a given column C
 COLMAX(R,C)    maximum value over the columns for a given row R
 NASHEQ(R,C)    matrix of 0-1 where 1 is a Nash equilibrium
 PROFHNE(R,C)   profit of firm h in Nash equilibrium
 PROFFNE(R,C)   profit of firm f in Nash equilibrium;

* small maintenance costs -0.1 when not entering, not needed



C:\jim\COURSES\8858\code-bk 2012\M4-2.gms  Monday, January 09, 2012 4:51:12 AM Page 2

TABLE PAYOFFH(*,*)
      SF0   SF1   SF2
SH0   -.1   -.1   -.1
SH1    10     6     3
SH2    12     5     2 ;

TABLE PAYOFFF(*,*)
      SF0   SF1   SF2
SH0   -.1    10    12
SH1   -.1     6     5
SH2   -.1     3     2 ;

DISPLAY PAYOFFH, PAYOFFF;

ROWMAX(R,C) = 1$(PAYOFFH(R,C) EQ SMAX(RR, PAYOFFH(RR,C)));
COLMAX(R,C) = 1$(PAYOFFF(R,C) EQ SMAX(CC, PAYOFFF(R,CC)));

DISPLAY ROWMAX, COLMAX;

NASHEQ(R,C) = ROWMAX(R,C)*COLMAX(R,C);

DISPLAY NASHEQ;

PROFHNE(R,C) = PAYOFFH(R,C)$NASHEQ(R,C);

PROFFNE(R,C) = PAYOFFF(R,C)$NASHEQ(R,C);

DISPLAY PROFHNE, PROFFNE;

*CASE 2: MARKETS TOO SMALL FOR A FIRM TO ENTER AGAINST A TWO-PLANT RIVAL
*subtract 4 from each payoff strategies 1 and 2

TABLE PAYOFFH2(*,*)
      SF0   SF1   SF2
SH0   -.1   -.1   -.1
SH1     6     2    -1
SH2     8     1    -2 ;

TABLE PAYOFFF2(*,*)
      SF0   SF1   SF2
SH0   -.1     6     8
SH1   -.1     2     1
SH2   -.1    -1    -2 ;

ROWMAX(R,C) = 1$(PAYOFFH2(R,C) EQ SMAX(RR, PAYOFFH2(RR,C)));
COLMAX(R,C) = 1$(PAYOFFF2(R,C) EQ SMAX(CC, PAYOFFF2(R,CC)));

NASHEQ(R,C) = ROWMAX(R,C)*COLMAX(R,C);

DISPLAY NASHEQ;

PROFHNE(R,C) = PAYOFFH2(R,C)$NASHEQ(R,C);
PROFFNE(R,C) = PAYOFFF2(R,C)$NASHEQ(R,C);

DISPLAY PROFHNE, PROFFNE;

*CASE 3: LOWER FIRM FIXED COSTS, RAISE PLANT FIXED COSTS
*makes two-plant production more profitable
*add 2 when playing strategy 2

TABLE PAYOFFH3(*,*)
      SF0   SF1   SF2
SH0   -.1   -.1   -.1
SH1    10     6     3
SH2    14     7     4 ;

TABLE PAYOFFF3(*,*)
      SF0   SF1   SF2
SH0   -.1    10    14
SH1   -.1     6     7
SH2   -.1     3     4 ;

ROWMAX(R,C) = 1$(PAYOFFH3(R,C) EQ SMAX(RR, PAYOFFH3(RR,C)));
COLMAX(R,C) = 1$(PAYOFFF3(R,C) EQ SMAX(CC, PAYOFFF3(R,CC)));

NASHEQ(R,C) = ROWMAX(R,C)*COLMAX(R,C);

DISPLAY NASHEQ;

PROFHNE(R,C) = PAYOFFH3(R,C)$NASHEQ(R,C);
PROFFNE(R,C) = PAYOFFF3(R,C)$NASHEQ(R,C);

DISPLAY PROFHNE, PROFFNE;


